6. ADA GYAKORLAT

(HATÓKÖR, LÁTHATÓSÁG, ÉLETTARTAM, KIVÉTEL KEZELÉS)

(verem, getdiscreet, faktorialis, max_hely)

Példa - HF  

Írjunk egy eljárás sablont a TEXT_IO programcsomag PUT_LINE eljárását használva különböző típusú értékekek kiíratására. A sablonnak két paramétere legyen:

- T típus a kiírandó elem típusa (korlátozott private)

- toString nevű fgv., melynek T a bemenő paraméterének típusa és egy stringet ad vissza.

kiiratás: peldak / hf / kiir.ads, kir.adb, demo.adb 

-------------------------------------------------------------------------------------------------------------------------------------

KIVÉTELEK:

Kiváltódás, definiálás,  terjedés, lekezelés, kiváltás

KIVÁLTÓDÁS (Előre definiált kivételek):

· Constraint_Error (és Numeric_Error)

· Program_Error

· Storage_Error

· Tasking_Error

· A szabványos könyvtárak által definiált kivételek (pl. Ada.IO_Exceptions.End_Error)

DEFINIÁLÁS

Tele_A_Verem: Exception;

Baki: Exception_Id ; 
-- Exception Id lekérdezése: Consrtraint_Error'Identity
Baki_Példány: Exception_Occurrence

TERJEDÉS:

· Egy utasítás végrehajtása közben váltódhat ki („fellép”)

· A hívottból a hívóba terjed. A hívottban is fellép a hívás helyén.

· A végrehajtási verem mentén obáljuk ki az aktivációs rekordokat.

·  Ha a verem kiürül, leáll a program.

· Alprogramok

· „Dinamikus tartalmazás”

· Ha egy meghívott alprogramban fellép és nem kezeljük le, akkor fellép a hívó blokkban is a hívás helyszínén

· Blokk utasítás

· „Statikus tartalmazás”

·  Ha egy blokk utasításban fellép és nem kezeljük le, akkor fellép a tartalmazó blokkban is a tartalmazás helyszínén
KIVÉTELEK KEZELÉSE:

Ada: blokk kivételkezelő része

· alprogram és blokk utasítás is lehet

· a végén opcionális kivételkezelő rész

· benne kivételkezelő ágak

procedure A ( S: in out Stack; N: in out Natural ) is



X: Positive; 

begin



Push(S, N);



Pop(S, X);
exception


when Constraint_Error => Push(S,N);






       N := 1;



when Tele_A_Verem => null;


   when others => ; -- az összes maradékot (a terjedés által ide kerülteket is 

lekezeli) 

end A;

Ada.Exceptions csomag: Függvények és típusok a kivételek kezelésére
KIVÉTELEK KIVÁLTÁSA

· Lehet:

· deklarációs részben (kezelés az őt dinamikusan tartalamazó legszükebb blokkban)

· kivétel kezelő részben (kezelés az őt dinamikusan tartalamazó legszükebb blokkban)

· utasítás részben (kezelés magában az utasítás blokkban)

raise Tele_A_Verem;

Raise_Exception(Baki);

Raise_Exception(Baki,"szöveges üzenet");

-- új kivétel példányok jönnek létre

Reraise_Occurrence(Baki_Példány); -- nem jön létre új kivétel példány

Kivételek használata: 

· hibajelzés (Üres veremből adat nem vehető ki)

· biztonságos menekülés (adatokat menteni, file-kat bezárni)

· megmentés (ne álljon le a program) 

· vezérlési szerkezet

Példa1.   vermek -- kivétel kiváltása utasítás kezelő részben és kivétel kezelő részben, kivétel kezelése, Hibajelzés kivétellel 

Valósítsd meg a verem típust sablon csomagban a szokásos műveletekkel! 

A sablon paramétere legyen:

- az elemek típusa, mint átlátszatlan (private)  típus, 

Maga verem legyen korlátozott átlátszatlan, amit diszkriminánsos rekord segítségével valósíts meg (a verem hossza a diszkrimináns). A rekord mezői pedig legyenek egy tömb, ami az elemeket tartalmazza, és egy mutató (veremtető), ami a verem tetejére mutat. Kezdetben a

verem legyen üres, minden újonnan deklarált változó esetén. 

A szokásos (Push, Pop, Top) műveletek mellett legyen még egy-egy függvényünk a verem ürességének illetve telítettségének vizsgálatára, valamint a verem aktuális méretének kiszámolására.!

A Push, Top és Pop műveletekben váltsuk ki az Üres_A_Verem ill. a Tele_A_Verem kivételeket a megfelelő esetekben. 

vermek.ads

generic

      type Elem is private;

package Vermek is

      type Verem (Max: Positive) is limited private;


      Üres_A_Verem, Tele_A_Verem: exception; -- !!!!!!

      procedure Push( V: in out Verem; E: in Elem );

         -- kiválthat Tele_A_Verem kivételt

      procedure Pop( V: in out Verem; E: out Elem );

         -- kiválthat Üres_A_Verem kivételt

      function Top( V: Verem ) return Elem;

         -- kiválthat Üres_A_Verem kivételt

      function Is_Empty( V: Verem ) return Boolean;

      function Is_Full( V: Verem ) return Boolean;

      function Size( V: Verem ) return Natural;

private

      type Tömb is array( Integer range <> ) of Elem;

      type Verem(Max: Positive ) is record

                                        Adatok: Tömb(1..Max);

                                        Veremtető: Natural := 0;

                                    end record;

end Vermek;

-------------------------------------------------------------------------------------------------------------

vermek.adb

package body Vermek is

      procedure Push( V: in out Verem; E: in Elem ) is

      begin


          if Is_Full(V) then

              raise Tele_A_Verem; --kivétel kiváltása  utasítássorozatban , 

         else

              V.Veremtető := V.Veremtető + 1;

              V.Adatok(V.Veremtető) := E;

          end if;

      end;

      procedure Pop( V: in out Verem; E: out Elem ) is

      begin

          E := Top(V);

          V.Veremtető := V.Veremtető - 1;

      end;

      function Top( V: Verem ) return Elem is

      begin

          return V.Adatok(V.Veremtető);


       exception

          when Constraint_Error => raise Üres_A_Verem;

--kivétel kiváltása kivétel kezelő részben 

    end;

      function Is_Empty( V: Verem ) return Boolean is

      begin

          return V.Veremtető = 0;

      end;

      function Is_Full( V: Verem ) return Boolean is

      begin

          return V.Veremtető >= V.Max;

      end;

      function Size( V: Verem ) return Natural is

      begin

          return V.Veremtető;

      end;

end Vermek;

-------------------------------------------------------------------------------------------------------------

menu.adb

Írjunk demo programot, amely egy ciklikus menü segítségével lehetővé teszi a felhasználonak egy általa megadott méretű verembe elemek betételét és kivételét. Itt kezeljük le a verem sablonban fellépett kivételeket, úgy, hogy szövegesen írjuk ki a kivétel okát  (Tele volt a verem stb,)

with Vermek, Ada.Text_IO, Ada.Integer_Text_IO;

use Ada.Text_IO, Ada.Integer_Text_IO;

procedure Menu is

    package Int_Vermek is new Vermek(Integer); use Int_Vermek;

    Max: Integer;

    procedure Menu_Ciklus ( V: in out Verem ) is

        Parancs: Integer;

        procedure Betesz( V: in out Verem ) is

             Adat: Integer;

        begin

             Put("Mit akar betenni? "); Get(Adat);

             Push(V,Adat);

        exception

             when Tele_A_Verem =>

                     Put_Line("Nem tudom végrehajtani. Már megtelt a verem.");

 end Betesz;

        procedure Kivesz( V: in out Verem ) is

             Adat: Integer;

        begin

             Pop(V,Adat);

             Put("A kivett adat: "); Put(Adat); New_Line;


        exception

             when Üres_A_Verem =>

                     Put_Line("Nem tudom végrehajtani. Üres a verem.");

        end Kivesz;

    begin

        loop

            Put_Line("Elvégezhető tevékenységek: push (1), pop (2), quit (0)");

            Get(Parancs);

            case Parancs is

                when 1 => Betesz(V);

                when 2 => Kivesz(V);

                when 0 => exit;

                when others => null;

            end case;

        end loop;

    end Menu_Ciklus;

begin

    Put( "Mekkora vermet szeretne használni? " ); Get(Max);

    declare

        V: Verem(Max);

    begin

        Menu_Ciklus(V);

    end;

end Menu;

-------------------------------------------------------------------------------------------------------------

Skip_Line

· Skip_Line causes remainder of current input line to be ignored by later calls to get

        -- Demonstrates how skip_line moves to next line, 

        -- and ignores the rest of a line

        with Ada.Text_IO; use Ada.Text_IO;

        procedure doStrings1 is

           s: String(1 .. 2);

           t: String(1 .. 3);

        begin

           put("Enter 2 or more characters and press return: ");

           get(s);

           put_line("first 2: " & s);

           new_line;

           skip_line;

           put("Enter 3 or more characters and press return: ");

           get(t);

           put_line("next 3: " & t);

        end doStrings1;

SAMPLE RUN:

Enter 2 or more characters and press return: abcde

first 2: ab

Enter 3 or more characters and press return: PQRST

next 3: PQR

Character, Get/Put Loop, End_Of_Line

        -- Demonstrates how end_of_line tests for end of line

        -- and how skip_line moves to next line

        with Ada.Text_IO; use Ada.Text_IO;

        procedure doStrings2 is

           c: Character;

           s: String(1 .. 80);           

           length: Natural;

        begin

           put("Enter some text and press enter: ");

           while (not Ada.Text_IO.End_Of_Line) loop

              get(c);

              put(c);

           end loop;

           skip_line;  -- move to next line of input

                       --   since we use get_line next

           new_line;

           put("Enter another line: ");

           get_line(s, length);

           put_Line(s);

        end doStrings2;

SAMPLE RUN:

Enter some text and press enter: ab

ab

Enter another line: def

def

http://www.radford.edu/~nokie/classes/320/stringio.html

Példa2 GET_DISCREET --- KIVÉTEL KEZELÉS SABLONBAN, PROGRAM LEÁLLÁS MEGAKADÁLYOZÁSA
Írjunk függvény sablont egy diszkrét elem típussal paraméterezve, amely beolvas egy inputot egy stringbe, és visszadja a beolvasott értéket a sablon paraméterébe megadott elemtípusként. Ha nem alakítható át a stringbe beolvasott elem az adott  elem típusúvá (pl. integer-ré), akkor Constraint Error lépne fel, amit kezeljünk le egy kivétel kezelő részben még a sablonon belül. 

get_discreet.ads

generic


type Element is (<>);

function Get_Discreet return Element;

-------------------------------------------------------------------------------------------------------------

get_discreet.adb

with TEXT_IO; use TEXT_IO;

function Get_Discreet return Element is

   Str: String(1..Element'Width+1);

   Len: Natural;

begin

   loop

      declare

      begin

          Get_Line(Str,Len);

          if Len <= Element'Width then

              return Element'Value( Str(1..Len) );

          else Skip_Line;

          end if;

      exception

          when Constraint_Error => null;

      end;

   end loop;

end Get_Discreet;

-------------------------------------------------------------------------------------------------------------

Írjunk demo programot a get_discreet sablon használatára integerek beolvasására.

proba.adb

with Get_Discreet, Text_IO; use Text_IO;

procedure Proba is

   function Get_Int is new Get_Discreet(Integer);

begin

   Put_Line( Integer'Image(Get_Int) );

end Proba;

-------------------------------------------------------------------------------------------------------------

Példa3 faktorialis -- KIVÉTELKEZELÉS  ÉS HASZNÁLATA VEZÉRLÉSI SZERKEZETKÉNT

Írjunk ciklus és elágazás nélkül, pusztán rekurziót és kivételkezelsét alkalmazva faktoriálist számító függvényt (Gyorsan ki lehet lépni egymásba ágyazott ciklusokból, de vigyázni kell, mert a kivétel kezelő mechanizmus erőforrás igényes, ezért ha lehet EXIT-tet alkalmazzunk.)

fact.adb

function Fact ( N: Natural ) return Positive is

begin

    return Fact(N-1) * N;

exception

    when Constraint_Error => return 1;

end Fact;

fact_demo.adb

with Fact, Ada.Integer_Text_IO;

use Ada.Integer_Text_IO;

procedure Fact_Demo is

   N: Natural;

begin

   Get(N);

   Put(Fact(N));

end;

--------------------------------------------------------------------------

Példa4 maxhely -- KIVÉTEL SABLON PARAMÉTEREKÉNT

Írjunk sablont egy vektor maximális helyének megkereséséhez. A sablon paraméterei legyenek: 

- a vektorelemek típusa (korlátozott átlátszatlan),

- a vektorindexek típusa (diszkrét),

- a vektor típusa,

- egy "<" függvény, amely az összehasonlítást végzi

- Egy EXCEPTION_ID típusú kivétel, kezdeti értékként a Constraint_Error Id-jével

    inicializálva, amely 0 hosszúságú tömb esetén váltódjon ki

A Max_Hely függvény bemenő paramétere legyen a tömb, és adja vissza a maximális elem helyének indexét. 

max_hely.ads

with Ada.Exceptions; use Ada.Exceptions;

generic


     type Elem is private; -- HIBA VAN, MERT A PROGRAMBAN LIMITED IS VAN, ÉS ÚGY NEM MŰKÖDIK

     type Index is (<>);


     type Tömb is array (Index range <>) of Elem;


     with function "<" ( A, B: Elem ) return Boolean is <>;


     Üres_Tömb: Exception_Id := Constraint_Error'Identity;

function Max_Hely ( T: Tömb ) return Index;

max_hely.adb

function Max_Hely ( T: Tömb ) return Index is

begin



if T'Length = 0 then




Raise_Exception( Üres_Tömb ); --, "Üres a tömb" );



else 
declare





Mh: Index := T'First;



        begin





for I in T'Range loop






if T(Mh) < T(I) then Mh := I; end if;





end loop;





return Mh;



        end; 





end if;

end Max_Hely;

-------------------------------------------------------------------------------------------

Példa5
Írjunk sablont egy vektor maximális elemének megkereséséhez. A sablon paraméterei legyenek: 

- a vektorelemek típusa (korlátozott átlátszatlan),

- a vektorindexek típusa (diszkrét),

- a vektor típusa,

- egy "<" függvény, amely az összehasonlítást végzi

- Egy EXCEPTION_ID típusú kivétel, kezdeti értékként a Constraint_Error Id-jével

    inicializálva, amely 0 hosszúságú tömb esetén váltódjon ki

A Max_Hely függvény bemenő paramétere legyen a tömb, és adja vissza a tömb maximális elemét. 

max.ads

with Ada.Exceptions; use Ada.Exceptions;

generic


     type Elem is private; -- HIBA VAN, MERT A PROGRAMBAN LIMITED IS VAN, ÉS ÚGY NEM MŰKÖDIK

     type Index is (<>);


     type Tömb is array (Index range <>) of Elem;


     with function "<" ( A, B: Elem ) return Boolean is <>;


     Üres_Tömb: Exception_Id := Constraint_Error ' Identity;

function Max ( T: Tömb ) return Elem;

max.adb

function Max ( T: Tömb ) return Elem is

begin



if T ' Length = 0 then




Raise_Exception( Üres_Tömb ); --, "Üres a tömb" );



else declare




Mh: Index := T ' First;



        begin





for I in T'Range loop






if T(Mh) < T(I) then Mh := I; end if;





end loop;





return T(Mh);



        end; 





end if;

end Max;

----------------------------------------------------------------------------------------------

Példa6
Írjunk demo eljárást a fenti két sablon használatára, úgy, hogy a 4 db sablon példány float típusó elemeket tartalmazó tömbök minimális ill. maximális elemeit és azok helyének indexeit adja vissza. A példányosításhoz szükséges egy Baki nevű exception típusú változó, amit a szükséges Exception_Id típusra a Baki'Identity attribútummal konvertálhatunk. Példaként egy üres tömböt is vegyünk fel, annak vizsgálatára, hogy eléri-e a Baki kivételt a program.

max_demo.adb

with Max, Max_Hely;

with Ada.Integer_Text_IO, Ada.Float_Text_IO;

use Ada.Integer_Text_IO, Ada.Float_Text_IO;

procedure Max_Demo is



type T is array (Integer range <>) of Float;



Baki: exception;



function Float_Max is new Max(Float,Integer,T);



function Float_Min is new Max( Float, Integer, T, ">",








Baki ' Identity );



function Float_Max_H is new Max_Hely(Float,Integer,T);



function Float_Min_H is new Max_Hely( Float, Integer, T, ">",








Baki ' Identity );



X: T(1..0);    -- üres tömb



Y: T(1..10) := (1.0,5.0,3.0,7.0,2.0,65.0,3.0,56.0,3.0,56.0);

begin



Put( Float_Max_H(Y) );



Put( Float_Max(Y) );



Put( Float_Min_H(Y) );



Put( Float_Min(Y) );



--Put( Float_Max_H(X) );



--Put( Float_Max(X) );



--Put( Float_Min_H(X) );



Put( Float_Min(X) );

end;

PéldaHF

· Házi feladat

· a 4. példában kezeljük le egy ismertető szöveg kiírásával a fellépő Baki kivételt. Használjuk az Ada.Exceptions csomag szükséges függvényeit a kivételhez az üzenet szöveg hozzárendeléséhez ill. az üzenet szöveg kiírásához.  

FŰGGELÉK

Input of Strings and Characters, EOL and EOF 



Output of Strings

         with Ada.Text_IO; use Ada.Text_IO;

         procedure io is

         begin

            put("Hello");

            put_line("There");

            put_line("Hi Mom!");

            new_line(2);

            put_line("ITEC 320" & " is FUN" & "!");

            set_col(3);

            put_line("I like Ada!");

         end io;

    OUTPUT:

    HelloThere

    Hi Mom!

    ITEC 320 is FUN!

      I like Ada!



Input a String

· get(s) always tries to read s'length characters 

· but, get(s) ignores newlines! 

        -- Demonstrates input of a string

        with Ada.Text_IO; use Ada.Text_IO;

        procedure doStrings1 is

           s: String(1 .. 2);           

           t: String(1 .. 3);          

        begin

           put("Enter 5 characters and press return: ");

           -- input exactly 2 characters

           get(s);

           put_line("first 2: " & s);

           -- input exactly 3 characters

           get(t);

           put_line("next 3: " & t);

           new_line;

           put("Enter 5 characters on several lines: ");

           -- input exactly 2 characters, spread over several lines

           get(s);

           put_line("first 2: " & s);

           -- input exactly 3 characters, spread over several lines

           get(t);

           put_line("next 3: " & t);

        end doStrings1;

        SAMPLE RUN:

        Enter 5 characters and press return: abc d

        first 2: ab

        next 3: c d

        Enter 5 characters on several lines:a

        b

        first 2: ab

        c

        d

        e

        next 3: cde



Get_Line

· What if the length of the input is not known? 

· Use Ada.Text_IO.Get_Line(s, len) 

· inputs one line of input (but no more than s'length characters) 

· len will contain the number of characters that were read

· s(1..len) will contain the characters

· Example: 

        with Ada.Text_IO; use Ada.Text_IO;

        with Ada.Integer_Text_IO; use Ada.Integer_Text_IO;

        procedure doStrings2 is

           s: String(1 .. 80);           

           len: Natural;

        begin

           put("Enter fewer than 80 characters: ");

           get_line(s, len);
           put("len = ");

           put(len, 0); 

           new_line;

           put_Line("You entered |" & s(1 .. len)  & "|");

           put_Line("All of s: |" & s & "|");

        end doStrings2;

SAMPLE RUN:

Enter fewer than 80 characters: cde

len = 3

You entered |cde|

All of s |cde$đ˙žő `¨|

· Get_line(s, l) fills s until it reaches the end of either s or the line. 

· Any unfilled locations in s have their previous contents!! 

· You must make sure you use a slice of s: s(1 .. len) 



More on Get_Line

· The newline at the end of the line is NOT read into s

· If the end of the line is encountered, then get_line automatically calls skip_line 

· If the line has exactly the number of characters in the string, skip_line is NOT called and must be called explicitly.

· Notice that the put_line outputs a slice of the string. What would happen if the whole string were output?

        with Ada.Text_IO; use Ada.Text_IO;

        procedure doStrings4 is

           s: String(1 .. 4);           

           len: Natural;

        begin

           for i in 1 .. 6 loop

              get_line(s, len);

              put_line(s(1 .. len));  -- Slice

              if len = 4 then

                 skip_line;

              end if;

           end loop;

        end doStrings4;

SAMPLE RUN:

abcdefg

abcd

pqrstuv

pqrs

1234

1234

mno

mno

abcd

abcd

abcde

abcd



End_Of_File

1. How to test for end of file: Ada.Text_IO.End_Of_File 

2. Example - sum up an unknown number of integers:

        -- Illustrates use of End_Of_File and integer get

        with Ada.Text_IO; use Ada.Text_IO;

        with Ada.Integer_Text_IO; use Ada.Integer_Text_IO;

        procedure eof4 is

           i: Integer;

           s: Integer := 0;

        begin

           while (not End_Of_File) loop

              get(i);

              s := s + i;

           end loop;

           put(s);

           new_line;

        end eof4;



Application of End_Of_File - Echo input

        -- Echoes input.

        -- Reads and writes one character at a time

        with Ada.Text_IO; use Ada.Text_IO;

        procedure eofloop3a is

           c: Character;

        begin

           while (not End_Of_File) loop

              get(c);

              put(c);

           end loop;

        end eofloop3b;



Echo input - Improved version

        -- Echoes input.

        -- Reads and writes one character at a time

        with Ada.Text_IO; use Ada.Text_IO;

        procedure eofloop3b is

           c: Character;

        begin

           while (not End_Of_File) loop

              get(c);

              put(c);

              -- Move output to new line as needed

              if End_Of_Line then

                  new_line;          

              end if;

           end loop;

        end eofloop3b;



Get_Immediate

· Get_Immediate can be used for unbuffered character input

· Get_Immediate reads a character without waiting for a newline

        -- Echoes input, using get_immediate.

        with Ada.Text_IO; use Ada.Text_IO;

        procedure eofloop4 is

           c: Character;

        begin

           loop

              get_immediate(c);

              put(c);

              exit when c = 'x';

           end loop;

        end eofloop4;



Look_Ahead

· Ada.Text_IO.look_ahead allows the user to view the next character in the input without consuming that character (ie without moving the input pointer)

        with Ada.Characters.Handling; use Ada.Characters.Handling;

        with Ada.Text_IO; use Ada.Text_IO;

        with Ada.Integer_Text_IO; use Ada.Integer_Text_IO;

        procedure lookahead is

           c: Character;

           i: Integer;

           line: String(1 .. 80);

           len: Natural;

           EOL: Boolean;

        begin

           put("Enter a number or a string: ");

           look_ahead(c, EOL);

           put_line("The first character in the input: " & c);

           put_line("At end of line: " & Boolean'Image(EOL));

           new_line;

           if Ada.Characters.Handling.is_digit(c) then

              get(i);

              put("You entered this number: ");

              put(i);

              new_line;

           else

              get_line(line, len);

              put("You entered this string: ");

              put_line(line(1 .. len));

           end if;

        end lookahead;

SAMPLE RUN 1

Enter a number or a string: 123

The first character in the input: 1

At end of line: FALSE

You entered this number:         123

SAMPLE RUN 2

Enter a number or a string: abcde

The first character in the input: a

At end of line: FALSE

You entered this string: abcde

SAMPLE RUN 3

Enter a number or a string: 

The first character in the input: 

At end of line: TRUE

EOL is true.  You must have just hit enter

· Multiple calls to look ahead without intervening calls to get will all return the same value for C

· Second parameter returns whether or not input is currently at the end of line

· Value of C is unchanged if EOL is true

· You can't look several characters ahead

· Lookahead can be done at any point of the input



End_Of_File - Echo input using get_line

· Get_line can also be used to echo input, but there are some subtleties when the length of the input line is not less then the length of the string being used for input 

        -- Echoes its input.

        -- Demonstrates using multiple get_lines to input lines that are

        -- longer than 4 characters 

        with Ada.Text_IO; use Ada.Text_IO;

        procedure eofLoop2 is

           maxLength: Positive := 4;

           s: String(1 .. maxLength);           

           length: Natural;

        begin

           while (not End_Of_File) loop

              get_line(s, length);

              put(s(1 .. length));

              if length < maxLength then

                 new_line;

              elsif length = maxLength and End_Of_Line then

                 skip_line;

                 new_line;

              end if;

           end loop;

        end eofLoop2;

SAMPLE RUN: 

abc

abc

abcdef

abcdef

ghijklkm

ghijklkm

a

a

^D



Running Programs Interactively vs Using Redirection

· A program that reads from standard input can be run interactively or from a file using redirection

· Example: running eofloop3 (from above) interactively:

>eofloop3

line

line

another line

another line

^D

· In interactive programs on unix systems, entering ^D on a line by itself will cause end of file to return true. Enter ^D by holding down control and D at the same time.

· In interactive programs on Windows systems, entering ^Z <Enter> on a line by itself will cause end of file to return true. Enter this by holding down control and Z at the same time, and then pressing the Enter/Return key.

· To cause program eofloop3 to read from the file foo, use redirection:

>cat foo

line 1

line 2

>eofloop3 < foo

line 1

line 2

· Another use for redirection is to capture output:

>eofloop3 < foo > goo

>cat goo

line 1

line 2

· Note that the output redirection above only works if the file goo does not exist. If the output file exists, you can replace it with a new version using an exclamation point (ie !, aka bang) as follows:

>cat moo

line A

line B

>eofloop3 < moo >! goo

>cat goo

line A

line B



Skip_Line

· Skip_Line causes remainder of current input line to be ignored by later calls to get

        -- Demonstrates how skip_line moves to next line, 

        -- and ignores the rest of a line

        with Ada.Text_IO; use Ada.Text_IO;

        procedure doStrings1 is

           s: String(1 .. 2);

           t: String(1 .. 3);

        begin

           put("Enter 2 or more characters and press return: ");

           get(s);

           put_line("first 2: " & s);

           new_line;

           skip_line;

           put("Enter 3 or more characters and press return: ");

           get(t);

           put_line("next 3: " & t);

        end doStrings1;

SAMPLE RUN:

Enter 2 or more characters and press return: abcde

first 2: ab

Enter 3 or more characters and press return: PQRST

next 3: PQR



Character, Get/Put Loop, End_Of_Line

        -- Demonstrates how end_of_line tests for end of line

        -- and how skip_line moves to next line

        with Ada.Text_IO; use Ada.Text_IO;

        procedure doStrings2 is

           c: Character;

           s: String(1 .. 80);           

           length: Natural;

        begin

           put("Enter some text and press enter: ");

           while (not Ada.Text_IO.End_Of_Line) loop

              get(c);

              put(c);

           end loop;

           skip_line;  -- move to next line of input

                       --   since we use get_line next

           new_line;

           put("Enter another line: ");

           get_line(s, length);

           put_Line(s);

        end doStrings2;

SAMPLE RUN:

Enter some text and press enter: ab

ab

Enter another line: def

def



Summary of input routines

· Use these declarations:

   i: integer;

   c: character;

   s: String(1..80);

· Integer: get(i) will skip whitespace (including newlines) until it finds a digit, plus, or minus. It will input the following integer, convert it, and store it in i.

· Character: get(c) will input the next character (ie the character that the input pointer currently points to) and advance the input pointer. At the end of a line get will find the next character by skipping the newline and getting the first character on the next line. 

· String: get(s) will input the next 80 characters (actually it will input the number of characters specified as the length in the declaration - 80 in this case), skipping over any newlines that are encountered.

· String: get_line(s, len) will input up to 80 characters (actually up to the number of characters specified as the length to be in the declaration), up to the end of the line (but not including the newline character). The number of characters that were input is returned in len. 

If the line has fewer than 80 characters, then the input marker moves to the next line and the next input will read from that next line. If the line has 80 or more characters, then the input marker remains on the current line. 



An important WARNING on input from a file 

· Remember: get (of either a character or an integer) skips newlines 

· However, characters inserted at the end of a line differ by OS:

· Windows: newline and linefeed 

· Unix: newline only 

· Consequence: A get command compiled under unix can treat end of line in data files created under unix differently from data files created under windows.

· A get command compiled under unix that works correctly on data files created under unix may not work on data files created under windows.

· Bottom line: Create any data file on the same system that you used to compile a program that reads that file.



A final word on EOF in interactive programs 

· In interactive programs, get causes an End_Error if it reads the end of file character (ie control D or control Z).

· In interactive programs, an end of file test samples the input to see if it is the EOF character.

· In interactive programs, the end of file character is not sticky. 

· This means that if a program executes 2 EOF tests, then for both of them to be true, then two EOF characters must be entered.

· In a program that executes 2 EOF tests, it is possible to execute a regular get between them.

· This is weird and should be avoided if at all possible.

· When reading from files, EOF is sticky since the file has an actual end of file character (or since the length of the file is known and the operating system can determine when EOF is reached.

· When reading from files, get causes an End_Error if it does a read when the input pointer is at the end of the file.

2010. ősz, Ada beadandók

1. feladat


Leírás: Készítsen egy programot, amely segítségével el lehet tárolni a fájlrendszer aktuális könyvtárában lévő bejegyzéseket (fájlok, könyvtárak adatait). Az adatokat a program futása során billentyűzetről lehessen bevinni. A megvalósításhoz az alábbi adatszerkezeteket kell felhasználni.

Adott egy tömb, melyben variáns rekordokat tárolunk. A rekord diszkriminánsa egy felsorolásos típust vár (Type File_Type is (File, Directory); Vigyázz, a File_Type foglalt típusnév!).

Record tartalma:

Név              : String(1..100);

Hozzáférés_ideje : String(1..100);

Méret            : Natural; -- csak fájlok esetén

Jogosultság      : String(1..100); -- vagy Felsorolásos típus.


Kell egy iterátor, mely: hasNext, next metódusokat tartalmaz, és a könyvtár listázására használjuk.

A fenti követelményeken felül egy tesztprogram is kell készíteni, amely bemutatja a program működését.

2. feladat


Leírás: Készítsen programot a bináris keresés hatékonyságának bemutatására. Az adatokat billentyűzetről lehessen feltölteni, és az elemek a tömbbe rendezetten kerüljenek bele (pl. buborékrendezéssel). A program biztosítson lehetőséget a keresés bemutatására és a keresés végzett lépések számát írja ki. Az elemek típusa és a rajta értelmezett összehasonlító operátor a felhasználó által legyen definiálható.

A megvalósítás menete:

· Készítsünk egy sablon csomagot. 

· A csomag paramétere legyen egy elemtípus, valamint egy összehasonlító operátor: "<", továbbá az elemtípusból egy extremális érték 

· A csomag egy átlátszatlan típust valósítson meg úgy, hogy egy diszkriminánsos rekordban elhelyezünk egy tömböt, melynek méretét a diszkrimináns adja meg. 

· Valósítsa meg az Insert műveletet, ami rendezetten szúr be a tömbbe. 

· Valósítsa meg a FindEfficiency műveletet, ami bináris keresést hajt végre a tömbön, visszatérő értéke pedig Integer, amely megmondja, hogy hány lépést kellett végezni a bináris keresés során. Ha nem volt ilyen, akkor ez az érték legyen negatív. 

· A tömb az üres helyein az extremális értéket tárolja, amelyről feltesszük, hogy minden lehetséges beszúrható értéknél kisebb. 

· Készítsen egy egyszerű tesztkörnyezetet, amellyel billentyűzetről adatokat lehet beszúrni, illetve lekérdezéseket lehet végrehajtani. 

